
Supplement to article “Intermittent Faking of Personality Profiles in High-Stakes 

Assessments: A Grade of Membership Analysis” 

Example Mplus code for F-GOM analysis 

TITLE: Two-level mixture syntax for F-GOM model for 10 observed responses, 

assuming one factor at both Retrieve (ThetaR) and Select (ThetaS) stages. 

 

DATA:  

FILE=DataLONG.dat;  

 

VARIABLE:  

!data in long format; as many records per person as there are responses 

NAMES = personID scaleID x y; !y are responses and x are importance ratings 

USEVARIABLES = x y1-y10; 

WITHIN = x;  !item importance ratings is within-level covariate 

classes = c (2); !the latent class variable, with 2 levels 

CLUSTER=personID; 

 

DEFINE: 

! dummy variables assuming that ScaleID starts with 0 an goes to 9,  

! which is the default in DATA WIDETOLONG transformation 

DO (1 10) IF (ScaleID+1 EQ #) THEN y#=y ELSE y#=_MISSING; 

 

ANALYSIS: 

TYPE=TWOLEVEL MIXTURE;   

STARTS=0;  !confirmatory model, user-supplied starting values 

PROCESSORS=4;  INTEGRATION=9; !optional, to speed up estimation 

 

MODEL: 

%WITHIN% 

%OVERALL% 

c ON x;   !regression of decisions to edit on x as per equation (7) 

[c#1*0];  !starting value for the intercept alpha from equation (7) 

%c#1% !’real’ class 

y1-y10*4;  !class-specific res. variances 

%c#2% !’ideal’ class 

y1-y10*1;  !class-specific res. variances 

 

%BETWEEN% 

%OVERALL% 

!default model for Retrieve stage 

ThetaR BY y1-y10*1;  ThetaR@1; [ThetaR@0]; 

!default model for Select stage 

ThetaS BY y1-y10*;  ThetaS@1; [ThetaS@0]; 

ThetaR WITH ThetaS; 

 

!c#1 at between level is person’s propensity to respond in class #1 

!which is the ‘real’ class; therefore c#1 is reversed ThetaE 

c#1*1;   !starting value for variance of propensities c#1 

ThetaR ThetaS WITH c#1*0; !declare factors correlated with c#1 

 

%c#1%  !’real’ class; Retrieve model applies 

ThetaS BY y1-y10@0; !honest responses DO NOT indicate ThetaS 

[y1-y10*0];  !class-specific intercepts with starting values 

 

%c#2% !’ideal’ class; Select model applies 

ThetaR BY y1-y10@0; !faked responses DO NOT indicate ThetaR 

[y1-y10*2];   !class-specific intercepts with starting values 

 

PLOT:  

TYPE=PLOT3; 

SERIES=y1(1) y2-y10(*);  !plotting class-specific means 


